Représentation graphique et résolution d'équation

I Représentation graphique

I.1 Bibliothèque d'instructions graphiques

Par défaut, le logiciel Maple ne charge pas toutes les commandes dont il dispose ; il y en a beaucoup trop. Pour pouvoir utiliser les autres instructions, il est nécessaire de les charger en mémoire. C'est le cas de plusieurs instructions graphiques que nous allons utiliser dans ce cours.

```
> with(plots) ;
Warning, the name changecoords has been redefined
```

[*Interactive, animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, cylinderplot, densityplot, display, display3d, fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra_supported, polyhedraplot, replot, rootlocus, semilogplot, setoptions, setoptions3d, spacecurve, sparsematrixplot, sphereplot, surfdata, textplot, textplot3d, tubeplot*]

Par cette commande, nous venons de mettre en mémoire la bibliothèque (ou librairie) `plots` qui contient l'ensemble des instructions ci-dessus.

Remarque I.1. *La commande* `restart` *effacera de la mémoire à la fois le contenu des variables mais également les instructions que vous avez chargées.*

I.2 Graphe d'une fonction

I.2.a A partir de son expression

Nous représentons le graphe de la fonction f définie par : $f(x) = \mathrm{e}^{-\frac{x}{4}}\sin(x)$ entre 0 et 6π.

```
> plot(exp(-x/4)*sin(x),x=0..6*Pi) ;
```



On peut également représenter plusieurs courbes sur un même graphique. Il suffit de remplacer l'expression par la liste des expressions des fonctions (*i.e.* les expressions séparées par des virgules et cernées par des crochets) que l'on veut représenter :

```
> plot([exp(-x/4)*sin(x),exp(-x/4),-exp(-x/4)],x=0..6*Pi) ;
```



I.2.b A partir d'une fonction

Créons la fonction f définie par : $f(x) = 3\sin(x) + 2\cos(5x)$:

```
>  f :=x->3*sin(x)+2*cos(5*x) ;
```

$$f := x \to 3\sin(x) + 2\cos(5x)$$

Nous représentons cette fonction sur $[-2\pi, 2\pi]$:

```
>  plot(f,-2*Pi..2*Pi) ;
```



De manière similaire au cas précédent, nous pouvons représenter deux fonctions :

```
>  g :=x->3*sin(x) ;
   plot([f,g],-2*Pi..2*Pi) ;
```

$$g := x \to 3\sin(x)$$



I.3 Courbe paramétrée

Définition I.1 (Courbe paramétrée)**.** *Soit x et y deux fonctions à valeurs réelles définie sur un même intervalle I. L'ensemble $\mathcal{C}$ des points $M(t)$ de coordonnées $(x(t), y(t))$ pour $t \in I$ est appelé une courbe paramétrée.*

Remarque I.2. *Si x et y représentent la position en fonction du temps d'un mobile dans un plan, alors la courbe $\mathcal{C}$ est la trajectoire de ce mobile.*

Nous représentons la courbe paramétrée par : $x(t) = t + \sin(t)$ et $y(t) = 1 + \cos(t)$ pour $t \in [-\pi, 3\pi]$:

```
>  plot([t+sin(t),1+cos(t),t=-Pi..3*Pi]) ;
```



On peut également tracer une courbe en ne spécifiant que les fonctions :

```
>  plot([3*sin,cos,0..2*Pi]) ;
```



I.4 Courbe polaire

Définition I.2 (coordonnées polaires)**.** *Tout point M du plan admet des coordonnées de la forme :*

$$M(r\cos(\theta), r\sin(\theta)) \text{ où } (r,\theta) \in \mathbb{R}^2$$

Le couple (r,θ) est appelé des coordonnées polaires de M (elle ne sont pas uniques).

Remarque I.3. *Si $M(x,y)$, on peut toujours prendre : $r = \sqrt{x^2+y^2}$ et $\theta = (\vec{i}, \overrightarrow{OM})$.*

Définition I.3 (Courbe polaire)**.** *Soit r une fonction définie sur un intervalle I.*
L'ensemble $\mathcal{C}$ des points $M(\theta)$ de coordonnées polaires $(r(\theta),\theta)$ pour $\theta \in I$ est appelé une courbe polaire.

Nous traçons la courbe polaire définie par $r = \cos(3\theta)$ pour $\theta \in [0, 2\pi]$:

```
>   plot(cos(3*theta),theta=0..Pi,coords=polar) ;
```



I.5 Courbe définie par une équation cartésienne

Nous traçons l'ensemble d'équation $y^3 + x^2 - x - y = 0$ pour $x \in [-5, 5]$ et $y \in [-5, 5]$:

```
>   implicitplot(y∧3+x∧2-x-y=0,x=-5..5,y=-5..5) ;
```



I.6 Regrouper plusieurs courbes

Il peut être nécessaire de tracer sur un même graphique plusieurs courbes préalablement calculées. Pour cela, nous stockons chaque courbe dans des variables (ici, `C1, C2, C3`) et nous utilisons la commande `display` comme suit :

```
>   C1 :=plot([sin(t),cos(t),t=0..2*Pi]) :
    C2 :=plot(cos(3*theta),theta=0..Pi,coords=polar) :
    C3 :=implicitplot(x∧2+2*x+y∧3-y=0,x=-1..1,y=-1..1) :
    display([C1,C2,C3]) ;
```



Remarque I.4. *Lors de l'affectation des variables* `C1`*,* `C2` *et* `C3`*, il est préférable de ne pas afficher le résultat en utilisant le séparateur " :" au lieu de ";" car sinon Maple affiche la liste des coordonnées des points de la courbe ; leur valeur n'a a priori aucun intérêt.*

I.7 Quelques options

I.7.a Changement d'échelle

Lorsque l'on trace le graphe d'une fonction, nous spécifions dans quel intervalle varie les coordonnées x et y :

```
> plot(x^2,x=-5..5,y=-1..10) ;
```



Dans tous les cas, on peut le faire avec l'option `view` :

```
> plot(1+cos(3*theta),theta=0..2*Pi,coords=polar,view=[-1.5..2.5,-2..2]) ;
```



Nous pouvons imposer que le repère soit orthonormal grâce à l'option `scaling=constrained` :

```
> plot([t+sin(t),1+cos(t),t=-Pi..3*Pi],scaling=constrained) ;
```



I.7.b Couleurs et épaisseurs

On spécifie la couleur de la courbe grâce à l'option `color` :

```
> plot(x^2,x=-5..5,color=blue) ;
```



Remarque I.5. *Quelques couleurs disponibles sur Maple sont : black, blue, green, orange, red, white, yellow ...*

On spécifie l'épaisseur du trait grâce à l'option `thickness` :

```
> plot(x^3,x=-5..5,thickness=2) ;
```



Dans le cas où l'on trace plusieurs courbes, on met la liste des couleurs et des épaisseurs entre crochets :

```
> plot([x^2,x^3,x^4],x=-5..5,y=-10..10,color=[blue,orange,green],thickness=[1,2,1]) ;
```



Résumé :	`plot(Expr,var=a..b)`	trace le graphe de l'expression `Expr` en fonction de `var` entre `a` et `b`
	`plot(f,a..b)`	trace le graphe de la fonction `f` entre `a` et `b`
	plot([Expr$_1$, ..., Expr$_n$],var=a..b)	trace les graphes des expressions Expr$_1$, ..., Expr$_n$ entre `a` et `b` dans un même graphique.
	plot([f$_1$, ..., f$_n$],a..b)	trace les graphes des fonctions f$_1$, ..., f$_n$ entre `a` et `b` dans un même graphique.
	plot([Expr$_x$,Expr$_y$,t=a..b])	trace la courbe paramétrée par les expressions Expr$_x$ et Expr$_y$ entre `a` et `b`
	`plot(Expr,theta=a..b,coords=polar)`	Trace la courbe d'équation polaire `r=Expr` pour `theta` entre `a` et `b`.
	`implicitplot(Eq,x=a..b,y=c..d)`	trace la courbe d'équation `Eq` pour `x` entre `a` et `b` et pour `y` entre `c` et `d`
	display([C$_1$,...,C$_n$]) ;	Trace sur un même graphique les courbes C$_1$, ...,C$_n$
Options :	`y=a..b`	Restreint l'axe des abscisses à $[a, b]$.
	`scaling=constrained`	Repère orthonormé
	`view=[a..b,c..d]`	restreint l'axe des abscisses à $[a, b]$ et l'axe des ordonnées à $[c, d]$
	`color=couleur`	trace la courbe en la couleur spécifiée
	`thickness=n`	trace la courbe avec une épaisseur `n`

Exercice 1. *Étude d'une fonction*

On considère la fonction f définie sur $\mathbb{R}$ par :

$$f(x) = \frac{x^2 - x + 1}{\sqrt{x^2 + x + 1}}$$

1. Créer la fonction f.
2. (a) Déterminer $a_1 = \lim_{x \longrightarrow +\infty} \frac{f(x)}{x}$ puis : $b_1 = \lim_{x \longrightarrow +\infty} f(x) - a_1 x$.
   Ainsi la droite $\mathcal{D}_1$ d'équation $y = a_1 x + b_1$ est l'asymptote de f en $+\infty$.

   (b) Déterminer $a_2 = \lim_{x \longrightarrow +\infty} \frac{f(x)}{x}$ puis : $b_2 = \lim_{x \longrightarrow +\infty} f(x) - a_2 x$.
   Ainsi la droite $\mathcal{D}_2$ d'équation $y = a_2 x + b_2$ est l'asymptote de f en $-\infty$.
3. Calculer $\alpha = f(2)$ et $\beta = f'(2)$.
4. Tracer sur un même graphique le graphe de f, les asymptotes de f et la tangente de f en 2 sur l'intervalle $[-10, 10]$ en restreignant l'axe des ordonnées à $[0, 10]$. On fixera la couleur et l'épaisseur de chacune de ces courbes.

Exercice 2. *Une courbe paramétrée*

On considère la courbe Γ paramétrée pour $t \in \mathbb{R}$ par :

$$x(t) = \frac{t^3 - t}{1 + t^2} \quad \text{et} \quad y(t) = \frac{t^2}{1 + t^2}$$

1. Créer les fonctions x et y.
2. (a) Calculer $p_1 = \frac{y'(1)}{x'(1)}$.
   La droite $\mathcal{D}_1$ de pente p_1 passant par le point de coordonnées $(x(1), y(1))$ est la tangente de Γ au point de paramètre 1.

   (b) Calculer $p_2 = \frac{y'(-1)}{x'(-1)}$.
   La droite $\mathcal{D}_2$ de pente p_2 passant par le point de coordonnées $(x(-1), y(-1))$ est la tangente de Γ au point de paramètre -1.
3. (a) Calculer $\lim_{t \longrightarrow +\infty} x(t)$ et $\lim_{t \longrightarrow +\infty} y(t)$.

   (b) Calculer $\lim_{t \longrightarrow -\infty} x(t)$ et $\lim_{t \longrightarrow -\infty} y(t)$.

   Ces calculs montrent que la droite $\mathcal{D}$ d'équation $y = 1$ est l'asymptote de Γ lorsque t tend vers $+\infty$ et vers $-\infty$.
4. Tracer sur un même graphique la courbe Γ et les droites $\mathcal{D}_1$, $\mathcal{D}_2$ et $\mathcal{D}$ en restreignant l'axe des abscisses à $[-5, 5]$, l'axe des ordonnées à $[-2, 2]$ et en faisant varier t dans l'intervalle $[-10, 10]$. On fixera la couleur et l'épaisseur de chacune de ces courbes.

II Résolution d'équation

La fonction `solve` permet de résoudre de manière exacte des équations ou des système d'équations d'inconnues réelles ou complexes. Maple propose d'autres instructions pour résoudre les autres types d'équations (équations différentielles...). Il propose également des instructions pour faire du calcul approché de solutions. Nous étudierons ces instructions ultèrieurement.

II.1 Résolution avec une inconnue

Résolvons une première équation :

```
> solve(3*x+2=4*x+5,x) ;
```

$$-3$$

À l'issue de ce calcul, la variable x reste quelconque :

```
> x ;
```

$$x$$

La fonction renvoie donc la ou les solutions de l'équation sans modifier l'inconnue.

Résolvons d'autres équations :

```
> solve(x^2+x-2=0,x) ;
```

$$1,\ -2$$

```
> solve(x^2+6*x+9=0,x) ;
```

$$-3,\ -3$$

Remarque II.1. *Remarquons que Maple signale les racines doubles des équations polynomiales en les affichant deux fois.*

Nous pouvons également résoudre des équations ayant des paramètres :

```
> solve(m*x^2+(m+1)*x-2*m-1=0,x) ;
```

$$1,\ -\frac{2m+1}{m}$$

Remarque II.2. *La fonction* `solve` *ne traite pas les cas particuliers. Ici, elle ne traite pas le cas* $m = 0$.

Remarque II.3. *La fonction* `solve` *ne donne pas systématiquement toutes les solutions des équations :*

```
> solve(sin(x)=1/2,x) ;
```

$$\frac{\pi}{6}$$

Comment récupérer les solutions d'une équation ?

Si l'équation n'admet qu'une seule solution, il suffit de stocker le résultat dans une variable de son choix. Nous pouvons ensuite la manipuler comme bon nous semble :

```
> sol :=solve(4*x+1=5*x+3,x) ;
```

$$sol := -2$$

```
> sol ;
  sol^2 ;
```

$$-2$$
$$4$$

Dans le cas où l'équation admet deux solutions, on stocke le résultat dans une variable :

```
> S :=solve(exp(2*x)-6*exp(x)=-7,x) ;
```

$$S := \ln(3+\sqrt{2}), \ln(3-\sqrt{2})$$

Le type de `S` est noté `exprseq` :

```
> whattype(S) ;
```

$$exprseq$$

Nous appellerons ce type de variable des séquences. Ici, la séquence `S` contient la donnée des deux solutions de l'équation. La première solution de l'équation est `S[1]`et la seconde est `S[2]` :

```
> S[1] ;
  S[2] ;
```

$$\ln(3+\sqrt{2})$$
$$\ln(3-\sqrt{2})$$

Parfois, la fonction `solve` n'explicite pas les solutions de l'équation :

```
> S :=solve(x^4+x-1=0,x) ;
```

$$S := \mathrm{RootOf}(\_Z^4 + \_Z - 1, index = 1), \mathrm{RootOf}(\_Z^4 + \_Z - 1, index = 2),$$
$$\mathrm{RootOf}(\_Z^4 + \_Z - 1, index = 3), \mathrm{RootOf}(\_Z^4 + \_Z - 1, index = 4)$$

`RootOf` signifie "racine de...". Maple nous répond donc que les solutions de l'équation sont les racines numéros 1, 2, 3 et 4 du polynôme $Z^4 + Z - 1$.

La fonction `allvalues` permet d'expliciter ces racines lorsque cela est possible :

```
> allvalues([S]) ;
```

(Expression très compliquée)

Pour calculer seulement la première solution, on applique la commande `allvalues` au premier terme de la séquence `S` :

```
> allvalues(S[1]) ;
```

(Expression très compliquée)

II.2 Résolution d'un système d'équations avec une seule solution

Résolvons le système $\begin{cases} 3x+y=5 \\ 7x-2y=3 \end{cases}$:

```
> S :=solve({ 3*x+y=5 , 7*x-2*y=3 },{x,y}) ;
```

$$S := \{x = 1,\ y = 2\}$$

Les variables x et y restent des inconnues :

```
> x,y ;
```

$$x,\ y$$

Si nous voulons stocker les solutions dans x et y, nous utilisons la commande assign :

```
> assign(S) ;
  x,y ;
```

$$1,\ 2$$

La commande assign ne renvoie aucun résultat. Elle effectue les affectations correspondant aux égalités contenue dans S. Pour réinitialiser x et y, on utilise les commandes :

```
> x :='x' ;
  y :='y' ;
```

$$x := x$$
$$y := y$$

II.3 Résolution d'un système d'équations avec plusieurs solutions

Résolvons le système $\begin{cases} 3x^2 + y^2 = 7 \\ 7x^2 - 2y^2 = -1 \end{cases}$:

```
> S :=solve({ 3*x∧2+y∧2=7 , 7*x∧2-2*y∧2=-1 }, {x,y}) ;
```

$$S := \{y = 2,\ x = -1\}, \{x = 1,\ y = 2\}, \{y = -2, x = -1\}, \{x = 1, y = -2\}$$

Pour récupérer les résultats précédemment, nous pouvons combiner les techniques vues précédemment. Ainsi, S[3] contient la troisième solution. En appliquant l'instruction assign à S[3], nous effectuons les affectations correspondantes :

```
> S[3] ;
```

$$\{y = -2, x = -1\}$$

```
> assign(S[3]) ;
  x,y ;
```

$$-1,\ -2$$

Par la suite, nous ne pouvons plus utiliser la commande assign pour faire une affectation portant sur x et y :

```
> assign(S[2]) ;

  Error, (in assign) invalid arguments
```

Si nous voulons ensuite que les valeurs de x et y correspondent à la deuxième solution, il nous faut d'abord réinitialiser ces variables :

```
> x :='x' ;
  y :='y' ;
  assign(S[2]) ;
  x,y ;
```

$$x := x$$
$$y := y$$
$$1,\ 2$$

II.4 Résolution d'inéquation

La fonction solve permet de résoudre des inéquations :

```
> solve(x∧2-4<=0,x) ;
```

$$\texttt{RealRange}(-2, 2)$$

"RealRange" signifie : "intervalle réel". L'ensemble des solutions de l'inéquation $x^2 - 4 \leq 0$ est donc $[-2, 2]$.

```
> solve(x∧2-4>0,x) ;
```

$$\texttt{RealRange}(-\infty, \texttt{Open}(-2)), \texttt{RealRange}(\texttt{Open}(2), \infty)$$

"Open" spécifie que la borne correspondante est ouverte. Ainsi, l'ensemble des solutions de l'inéquation $x^2 - 4 > 0$ est $]-\infty, -2[\cup]2, +\infty[$.

Nous pouvons également ajouter une inéquation à une équation :

```
> solve({ x∧3+3*x∧2+2*x=0, x<>0 },x) ;
```

$$-1,\ -2$$

Le symbole $\neq$ est noté <> sous Maple. Les solutions de l'équation $x^3 + 3x^2 + 2x = 0$ pour $x \neq 0$ sont -1 et -2.

Résumé :	`solve(Eq,x) ;`	résout l'équation (ou l'inéquation) `Eq` avec pour inconnue x
	`solve({Eq`$_1$`,...,Eq`$_n$`},{var`$_1$`, ..., var`$_m$`})`	résout le système d'équations `Eq`$_1$`,...,Eq`$_n$ avec pour inconnues `var`$_1$`, ..., var`$_m$
	`S[n]`	renvoie le `n`$^{\text{ème}}$ terme de la séquence `S`
	`RootOf(Eq)`	racine de l'équation `Eq`
	`RootOf(Eq,index=n)`	`n`$^{\text{ème}}$ racine de l'équation `Eq`
	`allvalues(Expr)`	explicite les termes `RootOf(...)`
	`allvalues([S])` ou `allvalues({S})`	explicite les termes `RootOf(...)` de chacun des termes de `S`
	`assign({x`$_1$`=a`$_1$`,...,x`$_n$`=a`$_n$`})`	Affecte les variables `x`$_1$, ..., `x`$_n$ les valeurs `a`$_1$`, ..., a`$_n$

Exercice 3. *Résolution d'un système d'équations*

1. Résoudre le système : $\begin{cases} x^2 + y^2 = 5 \\ 3x^2 - y^2 = 3 \end{cases} \qquad (S)$
2. Expliciter toutes les solutions de (S).
3. Stocker les solutions du système (S) dans les variables `x1, y1, x2, y2,...`.

Exercice 4. *Étude d'une fonction*

On considère la fonction f définie sur $\mathbb{R}$ par :

$$f(x) = \frac{x^2 + 4x + 4}{\sqrt{x^2 + 1}}$$

1. Créer la fonction f.
2. (a) Déterminer la dérivée de f.
   (b) Préciser l'unique annulation x_0 de f' et étudier le signe de f'.
3. (a) Déterminer la dérivée seconde de f.
   (b) Préciser les deux annulations x_1 et x_2 de f''.
4. Tracer le graphe de f et les tangentes de f en x_0, x_1 et x_2.
5. Tracer pour chaque point x_i le graphe de f et sa tangente en x_i sur l'intervalle $[x_i - 1; x_i + 1]$.

Exercice 5. *Racine d'un polynôme*

1. Factoriser le polynôme $P = X^2 + 3X + 1$.
2. Déterminer les racines de P.
3. Factoriser P en ajoutant $\sqrt{5}$ en second argument de la fonction utilisée à la question 1.
   On pourra consulter l'aide de la fonction incriminée si nécessaire.
4. Faites de même avec le polynôme $Q = X^2 + X + 1$.